Error Sheet For Spinning Top Software

Adjastment of MPU6050_tockn Library

The MPU module used in our system did not behave exactly like a standard MPU6050. During initial testing with the Adafruit MPU6050 library, the sensor returned a non-standard WHO_AM_I identification value instead of the expected MPU6050 value, indicating that the module was most likely a compatible or non-original MPU variant. Because of this, the standard Adafruit library did not initialize the sensor reliably, so we switched to / modified the .mpu6050_tockn library in order to communicate with the module despite its non-standard identification behavior

After communication was established, the sensor data recorded in the CSV file showed saturation during fast spinning. This indicated that the default measurement ranges used by the library were too narrow for the spinning-top system. Therefore, the gyroscope full-scale range was changed from ±500 deg/sec to ±2000 deg/sec by writing 0x18 to the GYRO_CONFIG register instead of 0x08. Since this configuration changes the gyro sensitivity, the gyro scale factor was also updated from 65.5 to 16.4 LSB/(deg/sec). This correction was applied both in the gyro offset calibration function and in the update function, so that the calculated offsets, angular velocity values, and integrated .gyro angles would all use the same physical scale

In addition, the accelerometer full-scale range was changed from ±2g to ±4g by writing 0x08 to the ACCEL_CONFIG register instead of 0x00. Accordingly, the accelerometer scale factor in the update function was changed from 16384 to 8192 LSB/g. This ensures that the raw acceleration values are converted according to the actual configured accelerometer range, and that the acceleration-based angle estimates are calculated using the .correct physical scale

Overall, these changes were necessary because the default library configuration did not match the behavior and dynamic range required by our hardware. Updating both the register settings and the corresponding conversion factors prevented saturated or incorrectly scaled measurements and made the MPU data more suitable for the spinning-top system.

```cpp
void MPU6050::begin(){
  writeMPU6050(MPU6050_SMPLRT_DIV, 0x00);
  writeMPU6050(MPU6050_CONFIG, 0x00);

  //writeMPU6050(MPU6050_GYRO_CONFIG, 0x08); // ±500 deg/sec
  writeMPU6050(MPU6050_GYRO_CONFIG, 0x18); // ±2000 deg/sec

  // g = gravity acceleration
  //writeMPU6050(MPU6050_ACCEL_CONFIG, 0x00); // ±2[g]
  writeMPU6050(MPU6050_ACCEL_CONFIG, 0x08); // ±4[g]

  writeMPU6050(MPU6050_PWR_MGMT_1, 0x01);
  this->update();
  angleGyroX = 0;
  angleGyroY = 0;
  angleX = this->getAccAngleX();
  angleY = this->getAccAngleY();
  preInterval = millis();
}
```

```cpp
void MPU6050::calcGyroOffsets(bool console, uint16_t delayBefore, uint16_t delayAfter){
  float x = 0, y = 0, z = 0;
  int16_t rx, ry, rz;

  delay(delayBefore);
  if(console){
    Serial.println();
    Serial.println("========================================");
    Serial.println("Calculating gyro offsets");
    Serial.print("DO NOT MOVE MPU6050");
  }
  for(int i = 0; i < 3000; i++){
    if(console && i % 1000 == 0){
      Serial.print(".");
    }
    wire->beginTransmission(MPU6050_ADDR);
    wire->write(0x43);
    wire->endTransmission(false);
    wire->requestFrom((int)MPU6050_ADDR, 6);

    rx = wire->read() << 8 | wire->read();
    ry = wire->read() << 8 | wire->read();
    rz = wire->read() << 8 | wire->read();

    //x += ((float)rx) / 65.5;
    //y += ((float)ry) / 65.5;
    //z += ((float)rz) / 65.5;
    x += ((float)rx) / 16.4;
    y += ((float)ry) / 16.4;
    z += ((float)rz) / 16.4;
  }

  gyroXoffset = x / 3000;
  gyroYoffset = y / 3000;
  gyroZoffset = z / 3000;

  if(console){
    Serial.println();
    Serial.println("Done!");
    Serial.print("X : ");Serial.println(gyroXoffset);
    Serial.print("Y : ");Serial.println(gyroYoffset);
    Serial.print("Z : ");Serial.println(gyroZoffset);
    Serial.println("Program will start after 3 seconds");
    Serial.print("========================================");
    delay(delayAfter);
  }
}
```

```cpp
void MPU6050::update(){
  wire->beginTransmission(MPU6050_ADDR);
  wire->write(0x3B);
  wire->endTransmission(false);
  wire->requestFrom((int)MPU6050_ADDR, 14);

  rawAccX = wire->read() << 8 | wire->read();
  rawAccY = wire->read() << 8 | wire->read();
  rawAccZ = wire->read() << 8 | wire->read();
  rawTemp = wire->read() << 8 | wire->read();
  rawGyroX = wire->read() << 8 | wire->read();
  rawGyroY = wire->read() << 8 | wire->read();
  rawGyroZ = wire->read() << 8 | wire->read();

  temp = (rawTemp + 12412.0) / 340.0;

  //accX = ((float)rawAccX) / 16384.0;
  //accY = ((float)rawAccY) / 16384.0;
  //accZ = ((float)rawAccZ) / 16384.0;
  accX = ((float)rawAccX) / 8192.0;
  accY = ((float)rawAccY) / 8192.0;
  accZ = ((float)rawAccZ) / 8192.0;

  angleAccX = atan2(accY, accZ + abs(accX)) * 360 / 2.0 / PI;
  angleAccY = atan2(accX, accZ + abs(accY)) * 360 / -2.0 / PI;

  //gyroX = ((float)rawGyroX) / 65.5;
  //gyroY = ((float)rawGyroY) / 65.5;
  //gyroZ = ((float)rawGyroZ) / 65.5;
  gyroX = ((float)rawGyroX) / 16.4; // 16.4 LSB/(deg/sec)
  gyroY = ((float)rawGyroY) / 16.4; // 16.4 LSB/(deg/sec)
  gyroZ = ((float)rawGyroZ) / 16.4; // 16.4 LSB/(deg/sec)

  gyroX -= gyroXoffset;
  gyroY -= gyroYoffset;
  gyroZ -= gyroZoffset;

  interval = (millis() - preInterval) * 0.001;

  angleGyroX += gyroX * interval;
  angleGyroY += gyroY * interval;
  angleGyroZ += gyroZ * interval;

  angleX = (gyroCoef * (angleX + gyroX * interval)) + (accCoef * angleAccX);
  angleY = (gyroCoef * (angleY + gyroY * interval)) + (accCoef * angleAccY);
  angleZ = angleGyroZ;

  preInterval = millis();

}
```